
(Part 1 of this article discussed the overall approach, the input stage, gain stage, frequency-

shaping stages, noise module, and output stage) 

 

Simulation Results 

The simulation results of the macromodel should be compared with the real world device to 

verify its functionality. The example op amp used in this article is the EL5165, a high-speed 

current-feedback amplifier. Some SPICE simulation results are compared with the real test 

curve. (Note: you can find the EL5165 Macromodel Netlist below the About the Authors 

box.) 

 

The gain plots in Figure 9 show a remarkable similarity between our model on the left at a 

gain of two and measurements from the EL5165 on the right. The measurements from the 

device are taken at both unity gain and a gain of two. There is less than 1% error between the -

3-dB frequency of the two gain-of-two measurements. 

 

 
Figure 9: AC response for Gain=+2 of the macromodel (left) and EL5165 device (right). 

 

At a gain of three, the discrepancy increases a bit as shown in Figure 10. Now the error in cut-

off frequency is within 5%. 

 



 
Figure 10: AC response for Gain=+3 of the macromodel (left) and EL5165 device (right). 

 

Since the feedback resistor has an effect on bandwidth, we offer a comparison of frequency 

response for a range of resistors in Figure 11. The values of resistance are listed on the plot on 

the right that comes from the data sheet. The left plot is from our macromodel. Again, the 

matching of -3 dB frequency is within 5%. 

 

 
Figure11: Frequency response with different Rf for the model (left) and the device (right). 

 

After examining the small-signal response versus frequency, the next logical step is to inspect 

the large-signal response as shown in Figure 12. Here we look at the slew rate of the step 

response with a gain of two driving a load of 150 ohms from a dual 5-volt supply. While the 

time base (x-axis) is at different resolutions in the two plots, the slew rate of the macromodel 

shown on the left is 5000 V/us. This is approximately 5% from the slew rate on the output 

signal of the EL5165 on the right. 

 

 
Figure 12: Large-signal step response for the macromodel (left) 

 and from the device data sheet (right). 



 

After frequency response and slew rate, our next goal is to match the noise performance, 

Figure 13. We have chosen to demonstrate the voltage noise, so the model in Figure 6 is 

included, but not the current noise model from Figure 7. The voltage noise model exhibits the 

spectrum shown in the plot on the left and has a value of 2.15 HznV /  at 1 MHz. The 

datasheet curve for voltage noise is on the right and has a value of 2.1 HznV /  at 1 MHz. 

 

 
Figure 13: Input noise voltage versus frequency for macromodel (left) and device (right). 

 

Conclusion 
A truly comprehensive SPICE macromodel for a current-feedback amplifier is developed. This 

macromodel includes effects such as transfer response, accurate AC response, DC offset and 

voltage noise. It is convenient to use such a model and change the parameters to fit other 

current-feedback amplifiers. Several of Intersil‟s current-feedback amplifiers use the same 

model topology.  

EL5165 Macromodel Netlist 
.subckt  EL5165         3    2    7    4    6 

* 

*Input Stage 

C_Cin1         0 3  1.8p  

V_Vos         1 N4150175 1.5mVdc 

I_I1         8 4 DC 85uAdc 

I_I2         7 5 DC 85uAdc   

I_Ib1         7 1 DC 2uAdc 

I_Ib2         7 2 DC 2uAdc 

G_Gi1         8 10 8 10 0.001 

G_Gi2         5 9 5 9 0.001  

G_Gi3         12 4 12 4 0.01 

G_Gi4         7 11 7 11 0.01 

G_Gb1         7 2 1 16 0.0000001 

Q_Q1         7 1 8 Inpn 

Q_Q2         11 9 2 Inpn 

Q_Q3         4 1 5 Ipnp 

Q_Q4         12 10 2 Ipnp 

D_D9         19 11 DX  

D_D10         12 18 DX  

V_V7         7 19 1.5dc 

V_V8         18 4 1.5Vdc 

C_C1         9 7  0.03p  

C_C2         4 10  0.03p 

C_Cs1         2 7  0.25p 

C_Cs2         4 2  0.25p 

I_I3         7 4 DC 3mAdc  

*Gain stage  

C_C3         VV2 7  0.405p  

C_C4         4 VV2  0.405p 

D_D3         13 7 DX  

D_D4         4 14 DX 

G_G1         7 VV2 7 11 0.01 

G_G2         VV2 4 12 4 0.01 

R_R7         VV2 7  10meg  

R_R8         4 VV2  10meg 

R_R17         13 0 1G 

R_R18         14 0 1G  

V_V5         13 VV2 1.6Vdc 

V_V6         VV2 14 1.6Vdc 

*High-order poles 

E_E3         16 4 7 4 0.5 

C_C5         VV3 7  0.0032p 

C_C6         4 VV3  0.0032p  

C_C7         4 VV4  0.0009p 



C_C8         VV4 7  0.0009p  

C_C9         4 VV5  0.0001p 

C_C10         VV5 7  0.0001p 

R_R9         VV3 7  100k 

R_R10         4 VV3  100k 

R_R11         VV4 7  100k 

R_R12         4 VV4  100k 

R_R15         VV5 7  100k 

R_R16         4 VV5  100k  

G_G3         4 VV3 VV2 16 0.00001 

G_G4         7 VV3 VV2 16 0.00001 

G_G9         7 VV4 VV3 16 0.00001 

G_G10         4 VV4 VV3 16 0.00001 

G_G11         7 VV5 VV4 16 0.00001 

G_G12         4 VV5 VV4 16 0.00001    

*Output stage 

G_G5         15 4 6 VV5 0.0001  

G_G6         17 4 VV5 6 0.0001 

G_G7         7 6 7 VV5 -0.04 

G_G8         6 4 VV5 4 -0.04 

R_R13         6 7  25 

R_R14         4 6  25 

D_D5         7 15 DX  

D_D6         7 17 DX  

D_D7         4 15 DY 

D_D8         4 17 DY 

*Voltage noise 

E_EN         N4150175 3 101 103 1 

V_V15         102 0 0.5Vdc 

V_V16         104 0 0.5Vdc 

D_DN1         102 101 Iden  

D_DN2         104 103 Iden  

R_R21         0 101  55  

R_R22         0 103  55 

*Current noise  

R_R23         0 NI2  0.022 

R_R24         0 NI2  0.022   

R_R25         0 NI1  0.022 

R_R26         0 NI1  0.022    

G_Gn1         3 0 NI1 0 1 

G_Gn2         2 0 NI2 0 1 

* 

* Models 

* 

.model Ipnp pnp(is=1e-15 bf=1E9 VAF=65) 

.model Inpn npn(is=1e-15 bf=1E9 VAF=65) 



.model Iden d(kf=100e-14 af=1) 

.MODEL DY D(IS=1E-20 BV=50 Rs=1) 

.MODEL DX D(IS=1E-15 Rs=1) 

.ends EL5165 

 

 

 


